/*
 * ESP32-C6 温湿度显示项目
 * 硬件: ESP32-C6 WROOM1 + DHT11 + LCD1602(IIC)
 */

#include <Wire.h>
#include <LiquidCrystal_I2C.h>
#include <DHT.h>

// DHT11配置
#define DHTPIN 4          // DHT11数据引脚连接到GPIO4
#define DHTTYPE DHT11     // 传感器类型

// I2C配置 (ESP32-C6默认I2C引脚)
#define I2C_SDA 6         // SDA引脚
#define I2C_SCL 7         // SCL引脚

// LCD1602地址 (通常是0x27或0x3F)
#define LCD_ADDR 0x27
#define LCD_COLS 16
#define LCD_ROWS 2

// 初始化对象
DHT dht(DHTPIN, DHTTYPE);
LiquidCrystal_I2C lcd(LCD_ADDR, LCD_COLS, LCD_ROWS);

void setup() {
  // 初始化串口（用于调试）
  Serial.begin(115200);
  delay(1000);
  Serial.println("ESP32-C6 温湿度监测系统启动...");
  
  // 初始化I2C
  Wire.begin(I2C_SDA, I2C_SCL);
  
  // 初始化LCD
  lcd.init();
  lcd.backlight();
  lcd.clear();
  
  // 显示欢迎信息
  lcd.setCursor(0, 0);
  lcd.print("Temp & Humidity");
  lcd.setCursor(0, 1);
  lcd.print("Initializing...");
  
  // 初始化DHT11
  dht.begin();
  
  delay(2000);
  lcd.clear();
}

void loop() {
  // 读取温湿度数据
  float humidity = dht.readHumidity();
  float temperature = dht.readTemperature();
  
  // 检查读取是否成功
  if (isnan(humidity) || isnan(temperature)) {
    Serial.println("读取DHT11失败！");
    lcd.clear();
    lcd.setCursor(0, 0);
    lcd.print("Sensor Error!");
    lcd.setCursor(0, 1);
    lcd.print("Check DHT11");
    delay(2000);
    return;
  }
  
  // 串口输出（用于调试）
  Serial.print("湿度: ");
  Serial.print(humidity);
  Serial.print("%  温度: ");
  Serial.print(temperature);
  Serial.println("°C");
  
  // LCD显示
  lcd.clear();
  
  // 第一行显示温度
  lcd.setCursor(0, 0);
  lcd.print("Temp: ");
  lcd.print(temperature, 1);  // 显示1位小数
  lcd.print((char)223);        // 度数符号
  lcd.print("C");
  
  // 第二行显示湿度
  lcd.setCursor(0, 1);
  lcd.print("Humi: ");
  lcd.print(humidity, 1);      // 显示1位小数
  lcd.print("%");
  
  // 等待2秒后更新（DHT11建议至少2秒更新间隔）
  delay(2000);
}

